5. The LCALL instruction is a ___-byte instruction.

SECTION 3.3: TIME DELAY FOR VARIOUS 8051 CHIPS

In the last section we used the DELAY subroutine. In this section we discuss how to generate various time delays and calculate exact delays for the 8051 and DS89C4x0.

Machine cycle for the 8051

The CPU takes a certain number of clock cycles to execute an instruction. In the 8051 family, these clock cycles are referred to as *machine cycles*. Table A-1 provides the list of 8051 instructions and their machine cycles. To calculate a time delay, we use this list. In the 8051 family, the length of the machine cycle depends on the frequency of the crystal oscillator connected to the 8051 system. The crystal oscillator, along with on-chip circuitry, provide the clock source for the 8051 CPU (see Chapter 8). The frequency of the crystal connected to the 8051 family can vary from 4 MHz to 30 MHz, depending on the chip rating and manufacturer. Very often the 11.0592 MHz crystal oscillator is used to make the 8051-based system compatible with the serial port of the IBM PC (see Chapter 10). In the original 8051, one machine cycle lasts 12 oscillator periods. Therefore, to calculate the machine cycle for the 8051, we take 1/12 of the crystal frequency, then take its inverse, as shown in Example 3-13.

Example 3-13

The following shows crystal frequency for three different 8051-based systems. Find the period of the machine cycle in each case.

(a) 11.0592 MHz (b) 16 MHz (c) 20 MHz

Solution:

(a) 11.0592 MHz/12 = 921.6 kHz; machine cycle is 1/921.6 kHz = 1.085 µs (microsecond)

(b) 16 MHz/12 = 1.333 MHz; machine cycle (MC) = 1/1.333 MHz = 0.75 µs

(c) 20 MHz/12 = 1.66 MHz; MC = 1/1.66 MHz = 0.60 µs

Example 3-14

For an 8051 system of 11.0592 MHz, find how long it takes to execute each of the following instructions.

(a) `MOV R3,#55` (b) `DEC R3` (c) `DJNZ R2,target`
(d) `LJMP` (e) `SJMP` (f) `NOP` (no operation)
(g) `MUL AB`

Solution:

The machine cycle for a system of 11.0592 MHz is 1.085 µs as shown in Example 3-13. Table A-1 in Appendix A shows machine cycles for each of the above instructions. Therefore, we have:

Instruction	*Machine cycles*	*Time to execute*
(a) `MOV R3,#55`	1	1×1.085 µs = 1.085 µs
(b) `DEC R3`	1	1×1.085 µs = 1.085 µs
(c) `DJNZ R2,target`	2	2×1.085 µs = 2.17 µs
(d) `LJMP`	2	2×1.085 µs = 2.17 µs
(e) `SJMP`	2	2×1.085 µs = 2.17 µs
(f) `NOP`	1	1×1.085 µs = 1.085 µs
(g) `MUL AB`	4	4×1.085 µs = 4.34 µs

Delay calculation for 8051

As seen in the last section, a delay subroutine consists of two parts: (1) setting a counter, and (2) a loop. Most of the time delay is performed by the body of the loop, as shown in Example 3-15.

Example 3-15

Find the size of the delay in the following program, if the crystal frequency is 11.0592 MHz.

```
        MOV   A,#55H      ;load A with 55H
AGAIN:  MOV   P1,A        ;issue value in reg A to port 1
        ACALL DELAY       ;time delay
        CPL   A           ;complement reg A
        SJMP AGAIN        ;keep doing this indefinitely
;----Time delay
DELAY:  MOV   R3,#200     ;load R3 with 200
HERE:   DJNZ R3,HERE      ;stay here until R3 become 0
        RET               ;return to caller
```

Solution:

From Table A-1 in Appendix A, we have the following machine cycles for each instruction of the DELAY subroutine.

```
                          Machine Cycle
DELAY:  MOV   R3,#200          1
HERE:   DJNZ R3,HERE           2
        RET                    2
```

Therefore, we have a time delay of $[(200 \times 2) + 1 + 2] \times 1.085\ \mu s = 436.255\ \mu s$.

Very often we calculate the time delay based on the instructions inside the loop and ignore the clock cycles associated with the instructions outside the loop.

In Example 3-15, the largest value the R3 register can take is 255; therefore, one way to increase the delay is to use NOP instructions in the loop. NOP, which stands for "no operation," simply wastes time. This is shown in Example 3-16.

Loop inside loop delay

Another way to get a large delay is to use a loop inside a loop, which is also called a *nested loop*. See Example 3-17.

Example 3-16

For an 8051 system of 11.0592 MHz, find the time delay for the following subroutine:

```
                                    Machine Cycle
DELAY:    MOV   R3,#250                 1

HERE:     NOP                           1
          NOP                           1
          NOP                           1
          NOP                           1
          DJNZ  R3,HERE                 2

          RET                           2
```

Solution:

The time delay inside the HERE loop is $[250(1 + 1 + 1 + 1 + 2)] \times 1.085\ \mu s = 1500 \times 1.085\ \mu s = 1627.5\ \mu s$. Adding the two instructions outside the loop we have $1627.5\ \mu s + 3 \times 1.085\ \mu s = 1630.755\ \mu s$.

If machine cycle timing is critical to your system design, make sure that you check the manufacture's data sheets for the device specification. For example, the DS89C420 has 3 machine cycles instead of 2 machine cycles for the RET instruction.

Example 3-17

For a machine cycle of 1.085 µs, find the time delay in the following subroutine.

```
DELAY:                                  Machine Cycle
          MOV   R2,#200                     1
AGAIN:    MOV   R3,#250                     1
HERE:     NOP                               1
          NOP                               1
          DJNZ  R3,HERE                     2
          DJNZ  R2,AGAIN                    2
          RET                               2
```

For the HERE loop, we have $(4 \times 250) \times 1.085\ \mu s = 1085\ \mu s$. The AGAIN loop repeats the HERE loop 200 times; therefore, we have $200 \times 1085\ \mu s = 217000$, if we do not include the overhead. However, the instructions "`MOV R3,#250`" and "`DJNZ R2,AGAIN`" at the beginning and end of the AGAIN loop add $(3 \times 200 \times 1.085\ \mu s) = 651\ \mu s$ to the time delay. As a result we have $217000 + 651 = 217651\ \mu s = 217.651$ milliseconds for total time delay associated with the above DELAY subroutine. Notice that in the case of a nested loop, as in all other time delay loops, the time is approximate since we have ignored the first and last instructions in the subroutine.

Example 3-21

Write a program to toggle all the bits of P1 every 200 ms. Assume that the crystal fre-
quency is 11.0592 MHz, and that the system is using the AT89C51.

Solution:

```
;Tested for AT89C51 of 11.0592 MHz.

          MOV    A,#55H
AGAIN:    MOV    P1,A
          ACALL  DELAY
          CPL    A
          SJMP   AGAIN

;----Time delay
DELAY:    MOV    R5,#2
HERE1:    MOV    R4,#180
HERE2:    MOV    R3,#255
HERE3:    DJNZ   R3,HERE3
          DJNZ   R4,HERE2
          DJNZ   R5,HERE1
          RET
```

$2 \times 180 \times 255 \times 2$ MC $\times$ 1.085 μs = 199,206 μs